Maximum Likelihood Equal-Variances Ratio Test and Estimation of Moderator Effects On Correlation with SPSS

Michael Smithson
Department of Psychology, The Australian National University

(email: Michael.Smithson@anu.edu.au)

SPSS Nonlinear Regression syntax components

The model in Smithson (2012) can be run in SPSS under its Constrained Nonlinear Regression (CNLR) procedure. SPSS requires three main components for these models:

1. Formulas for the submodels,
2. A formula for the negative log-likelihood kernel, which is the loss-function to be minimized, and
3. Names and starting-values for the model parameters.

Formulas for the submodels and negative log-likelihood kernel

The '#DV' is the dependent variable and it also must be the variable listed in the CNLR subcommand. The submodels are identified by the red font in the generic code shown below. The first two are the "nuisance parameter" submodels for the means of X and Y. The remaining three are the submodels for the standard deviations and the correlation. The negative log-likelihood kernel formula is identified by the **bold** text.

```
MODEL PROGRAM MNX =  MNY =  DX0 =  DY0 =  DR0 =    .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX .
COMPUTE #MUY = MNY .
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0 .
COMPUTE #DY = DY0 .
COMPUTE #SX = EXP(#DX) .
COMPUTE #SY = EXP(#DY) .
COMPUTE #RHO = (EXP(DR0)-1)/(EXP(DR0)+1) .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_  = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

Moderator and HeV effects may be added to the submodels in a straightforward way. If our moderator variable is Z, then a moderator effect on σ_y with coefficient DY1 would be represented in the submodel for the standard deviation of Y as follows:

```
COMPUTE #DY = DY0 + DY1*Z .
```

If we also add a first-order moderation effect with coefficient DR1 then the submodel for the correlation would look like this:

```
COMPUTE #RHO = (EXP(DR0+DR1*Z)-1)/(EXP(DR0+DR1*Z)+1) .
```

Model parameters and starting-values

It is important to get good starting values for these models. If Z is categorical, the best way to generate starting-values for the parameters is to use method of moments estimation. There are functions for doing this in R, available from the same webpage linked to this document. Alternatively, you can do so using SPSS or a spreadsheet program such as Excel.

Extracting More from CNLR

In this section I will cover two additional features of CNLR: Saving predicted values, residuals, and gradient values; and obtaining bootstrap standard-error estimates for the coefficients.

Saving computed variables

Predicted values, residuals, gradient values, and loss-function values for all cases can be obtained by inserting the /SAVE subcommand before the /CRITERIA subcommand. For instance,

```
/SAVE PRED RESID DERIVATIVES
```

will save the predicted values, residuals, and gradient values (derivatives) to the working data-file. The saved variables may then be used in the usual diagnostic fashion. For instance, summing the derivatives for the model shows the gradient is near 0 at the solution for each parameter, thereby supporting the claim that the solution is the true optimum.

Obtaining bootstrap standard-error estimates

SPSS does not compute the Hessian at the solution, so we cannot obtain asymptotic standard-error estimates in the usual way that we can from SAS or R. However, SPSS does provide bootstrap estimates (the default is 100 samples). To obtain bootstrap estimates of the standard errors (and the corresponding confidence intervals and correlation matrix of the estimates), insert the following subcommand before the /CRITERIA subcommand:

```
/BOOTSTRAP = N
```

where N is the number of samples desired. Usually 2000-4000 samples suffice for accurate estimates. This may take some time for your computer to complete. The standard-error estimates, confidence intervals and correlations of parameter estimates for Example 5 are displayed in the output below, using 3000 bootstrap samples. Two different kinds of "95% confidence intervals" are displayed: Bootstrap intervals using the standard error estimates and intervals based on excluding the bottom and top 2.5% of the bootstrap distribution.

Equal variance ratios tests

The model syntax described above is readily adapted to an EVR test by comparing models. The null model is estimated by assigning the same term to the moderator effect of Z on σ_x and σ_y. The syntax fragment below does this via the coefficient labeled D1.

```
COMPUTE #DX = EXP(DX0 + D1*Z) .
COMPUTE #DY = EXP(DY0 + D1*Z) .
```

The alternative model assigns separate coefficients to Z in the DX and DY submodels:

```
COMPUTE #DX = EXP(DX0 + DX1*Z) .
COMPUTE #DY = EXP(DY0 + DY1*Z) .
```

Two-category moderator example

This is a two-category moderator example in which the null hypothesis of EVR is true. It uses an artificial data-set sampled from two bivariate normal distributions, associated with one category of a binary moderator variable Z which takes values -1 and +1. For the first moderator category $\sigma_x^2 = 1$ and $\sigma_y^2 = 2$, while for the second category $\sigma_x^2 = 4$ and $\sigma_y^2 = 8$. Thus, the population variance ratio in both moderator categories is 1/2. The population

covariances are 1 and the means are 0 for both categories. The data-file is eqdat.sav. There are 500 observations in each category. In the first category, s_x^2 = 0.883 and s_y^2 = 1.958, so s_x^2/s_y^2 = 0.451. In the second category, s_x^2 = 4.068 and s_y^2 = 7.515, so s_x^2/s_y^2 = 0.541.
First, we fit a saturated model permitting unequal variance ratios:

```
MODEL PROGRAM MNX =  .01 MNY =  0.001 DX0 = 0.01 DY0 = 0.36 DR0 = 1.0 DX1 =
0.01 DY1 = 0.01 DR1 = -0.3  .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX .
COMPUTE #MUY = MNY .
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0+DX1*Z  .
COMPUTE #DY = DY0+DY1*Z  .
COMPUTE #SX = EXP(#DX)  .
COMPUTE #SY = EXP(#DY)   .
COMPUTE #RHO = (EXP(DR0+DR1*Z)-1)/(EXP(DR0+DR1*Z)+1)   .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_ = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

The output is shown in the table below.

Iteration History[b]

Iteration Number[a]	Value of Loss Function	Parameter							
		MNX	MNY	DX0	DY0	DR0	DX1	DY1	DR1
0.1	4480.058	.010	.001	.010	.360	1.000	.010	.010	-.300
1.1	3842.750	.010	.001	.583	.888	.858	.689	.591	-.509
2.1	3831.191	.011	.007	.546	.855	.869	.713	.611	-.597
3.1	3799.447	.038	.082	.592	.830	.832	.532	.497	-1.625
4.1	3796.686	.039	.083	.586	.837	.827	.532	.487	-1.653
5.1	3795.954	.042	.081	.582	.837	.813	.529	.490	-1.659
6.1	3795.550	.035	.084	.580	.835	.802	.528	.490	-1.662
7.1	3795.377	.036	.082	.583	.830	.797	.531	.487	-1.663
8.1	3769.208	-.215	-.287	.533	.823	.759	.520	.482	-1.463
9.1	3745.695	-.163	-.215	.510	.816	.723	.520	.482	-1.350
10.1	3661.604	.046	.037	.345	.719	.736	.491	.428	-.818
11.1	3643.619	.051	.041	.293	.688	1.039	.392	.310	-.525
12.1	3632.989	.024	.021	.332	.708	.986	.400	.335	-.775
13.1	3629.085	.018	.023	.323	.684	1.033	.397	.339	-.786
14.1	3627.889	.018	.027	.316	.672	1.104	.378	.337	-.796
15.1	3627.857	.019	.026	.320	.671	1.097	.383	.335	-.790
16.1	3627.853	.019	.026	.318	.672	1.096	.382	.337	-.790
17.1	3627.853	.019	.026	.319	.672	1.097	.382	.336	-.791
18.1	3627.853	.019	.026	.319	.672	1.097	.382	.336	-.790

Derivatives are calculated numerically.

a. Major iteration number is displayed to the left of the decimal, and minor iteration number is to the right of the decimal.

b. Run stopped after 18 iterations. Optimal solution is found.

The next model restricts the ratios to be equal:

```
MODEL PROGRAM MNX =  .01 MNY =  0.001 MX1 = .01 MY1 = 0.1 DX0 = 0.01 DY0 =
0.36 DR0 = 1.0 D1 = 0.01 DR1 = -0.3  .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX .
COMPUTE #MUY = MNY .
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0+D1*Z  .
COMPUTE #DY = DY0+D1*Z  .
COMPUTE #SX = EXP(#DX)  .
COMPUTE #SY = EXP(#DY)   .
COMPUTE #RHO = (EXP(DR0+DR1*Z)-1)/(EXP(DR0+DR1*Z)+1)  .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_  = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

The last line of the output is:

	-LL	MNX	MNY	DX0	DY0	DR0	D1	DR1
14.1	3629.304	.019	.027	.311	.680	1.096	.360	-.790

The unequal ratio model has -LL = 3627.853.
The chi-square difference is 2*(3629.304-3627.853) = 2.902, fairly close to the corresponding Mplus (v. 6.12) SEM chi-square value of 2.920.
The estimated variance ratio is
(exp(DX0 – DY0))^2 = (exp(-.369))^2 = 0.478,
agreeing with the Mplus estimate of .478.
Correlation estimates from the SPSS output:
r_1 = (exp(1.096-.790)-1)/(exp(1.096-.790)+1) = .152
r_2 = (exp(1.096+.790)-1)/(exp(1.096+.790)+1)= .737
The Mplus estimates are .151 and .737, so again agreement is very close.

<u>SEM example from Smithson (2012)</u>
The data-set for this example is as described in Smithson (2012), and the file is semex.sav.
Starting with SEMs for regression coefficients, we begin with a model that allows moderator effects on the variances and correlation with no restrictions on the parameters.

```
MODEL PROGRAM MNX =  .01 MNY =  0.001 DX0 = 0.01 DY0 = 0.36 DR0 = 1.0 DX1 =
0.01 DY1 = 0.01 DR1 = -0.3  .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX  .
COMPUTE #MUY = MNY .
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0+DX1*Z  .
COMPUTE #DY = DY0+DY1*Z  .
COMPUTE #SX = EXP(#DX)  .
COMPUTE #SY = EXP(#DY)   .
```

```
COMPUTE #RHO = (EXP(DR0+DR1*Z)-1)/(EXP(DR0+DR1*Z)+1)   .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_  = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

The last line of the output is:

	-LL	MNX	MNY	DX0	DY0	DR0	DX1	DY1	DR1
14.1	1840.682	.002	-.005	-.033	.381	.998	-.017	.317	.007

We now restrict the variance ratios to be equal:

```
MODEL PROGRAM MNX =  .01 MNY =  0.001 DX0 = 0.01 DY0 = 0.36 DR0 = 1.0 D1 =
0.01 DR1 = -0.3  .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX  .
COMPUTE #MUY = MNY .
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0+D1*Z  .
COMPUTE #DY = DY0+D1*Z  .
COMPUTE #SX = EXP(#DX)  .
COMPUTE #SY = EXP(#DY)   .
COMPUTE #RHO = (EXP(DR0+DR1*Z)-1)/(EXP(DR0+DR1*Z)+1)   .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_  = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

The last line of the output is:

	-LL	MNX	MNY	DX0	DY0	DR0	D1	DR1
13.1	1881.829	.007	-.020	-.006	.409	.936	.149	.006

The unequal ratio model has -LL = 1840.682.
The chi-square difference is 2*(1881.829-1840.682) = 82.294, quite close to Mplus' 82.246.
We can reject the EVR hypothesis.

We now estimate a model with HoV for *X* and moderator effects for the variance of *Y* and the correlation:

```
MODEL PROGRAM MNX =  .01 MNY =  0.001 DX0 = 0.01 DY0 = 0.36 DR0 = 1.0 DY1 =
0.01 DR1 = 0.1 .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX  .
COMPUTE #MUY = MNY .
```

```
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0 .
COMPUTE #DY = DY0+DY1*Z  .
COMPUTE #SX = EXP(#DX)  .
COMPUTE #SY = EXP(#DY)   .
COMPUTE #RHO = (EXP(DR0+DR1*Z)-1)/(EXP(DR0+DR1*Z)+1)  .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_  = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

The last line of the output is:

	-LL	MNX	MNY	DX0	DY0	DR0	DY1	DR1
12.1	1840.864	.003	-.005	-.033	.381	.998	.320	.023

The unrestricted model has -LL = 1840.682.
The chi-square difference is 2*(1840.864-1840.682) = 0.364, fairly close to Mplus' 0.370.
We retain the HoV for X hypothesis.

Finally, we estimate a model with HoV for X and equal correlations:

```
MODEL PROGRAM MNX =  .01 MNY =  0.001 DX0 = 0.01 DY0 = 0.36 DR0 = 1.0 DY1 =
0.01 .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX  .
COMPUTE #MUY = MNY .
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0 .
COMPUTE #DY = DY0+DY1*Z  .
COMPUTE #SX = EXP(#DX)  .
COMPUTE #SY = EXP(#DY)   .
COMPUTE #RHO = (EXP(DR0)-1)/(EXP(DR0)+1)  .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_  = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

The last line of the output is:

	-LL	MNX	MNY	DX0	DY0	DR0	DY1
11.1	1840.907	.003	-.005	-.033	.381	.998	.318

The HoV for X with unequal correlations model has -LL = 1840.864.

The chi-square difference is 2*(1840.907-1840.864) = 0.086, quite close to Mplus' 0.083, so we conclude that there is no moderation of correlations. The estimate of the correlation is
r = (exp(0.998)-1)/(exp(0.998)+1) = .461,
which is close to the Mplus estimate of .460.

Four-category moderator example

Now we consider a four-category moderator example, with EVR for the first two categories and for the second two, but not for both pairs of categories. An artificial data-set has been sampled from four bivariate normal distributions, associated with a four-category moderator variable *Z*, with 300 observations in each category. The variance ratio for the first two moderator categories equals 1/2, whereas for the third and fourth categories the ratio is 1/6. The first two categories' correlations also are identical. The data-set is named fourdat.sav. An appropriate design matrix for testing the four-groups model is shown below.

	σ_y dummy vars.			σ_x dummy vars.		
	z2	z3	z4	z2	z3	z4
Group 1	0	0	0	0	0	0
Group 2	1	0	0	1	0	0
Group 3	0	1	0	0	1	0
Group 4	0	1	1	0	1	1

The models to be compared are:

Mod1: $\log(\sigma_y) = \delta_{y0} + \delta_2 + \delta_{y3} + \delta_4$
$\log(\sigma_x) = \delta_{x0} + \delta_2 + \delta_{x3} + \delta_4$
$\text{logit}(\rho) = \delta_{r0} + \delta_{r3} + \delta_{r4}$

Mod2: $\log(\sigma_y) = \delta_{y0} + \delta_{y2} + \delta_{y3} + \delta_{y4}$
$\log(\sigma_x) = \delta_{x0} + \delta_{x2} + \delta_{x3} + \delta_{x4}$
$\text{logit}(\rho) = \delta_{r0} + \delta_{r3} + \delta_{r4}$

Mod3: $\log(\sigma_y) = \delta_{y0} + \delta_{y2} + \delta_{y3} + \delta_{y4}$
$\log(\sigma_x) = \delta_{x0} + \delta_{x2} + \delta_{x3} + \delta_{x4}$
$\text{logit}(\rho) = \delta_{r0} + \delta_{r2} + \delta_{r3} + \delta_{r4}$

Mod1 includes both EVR hypotheses and the equal-correlations hypothesis. Mod2 relaxes the EVR hypothesis and Mod3 relaxes both that and the equal-correlations hypothesis. Syntax for Mod1 is shown below. The other models' syntax can be developed by modifying this syntax in the obvious ways.

```
MODEL PROGRAM MNX =  .01 MNY =  0.001 DX0 = 0.03 DY0 = .4 DR0 = 1.8 D2 =
.18 DX3 = -.02 DY3 = -.5 DR3 = .5 D4 = 0.3 DR4 = -1.0  .
COMPUTE #DV = Y .
COMPUTE #MUX = MNX .
COMPUTE #MUY = MNY .
COMPUTE #DEVX = X - #MUX .
COMPUTE #DEVY = Y - #MUY .
COMPUTE #DX = DX0 + D2*Z2 + DX3*Z3 + D4*Z4 .
COMPUTE #DY = DY0 + D2*Z2 + DY3*Z3 + D4*Z4 .
COMPUTE #SX = EXP(#DX)  .
COMPUTE #SY = EXP(#DY)  .
COMPUTE #RHO = (EXP(DR0 + DR3*Z3 + DR4*Z4)-1)/(EXP(DR0 + DR3*Z3 +
DR4*Z4)+1)  .
COMPUTE PRED_ = #MUY + (X - #MUX)*#RHO*#SY/#SX .
```

```
COMPUTE RESID_ = #DV  - PRED_ .
COMPUTE LL = ((#SX**2)*(#DEVY**2) - 2*#RHO*#SX*#SY*#DEVX*#DEVY +
(#SY**2)*(#DEVX**2))/(2*(#RHO**2-1)*(#SX**2)*(#SY**2)) - LN(2*3.145927) -
0.5*(LN(1 - #RHO**2) + LN(#SX**2) + LN(#SY**2)) .
COMPUTE LOSS_  = -LL .
CNLR Y
  /OUTFILE='C:\SPSSFNLR.TMP'
  /PRED PRED_
  /LOSS LOSS_
  /CRITERIA STEPLIMIT 2 ISTEP 1E+20 .
```

The table below shows -LL and chi-square differences for the three models. The df differences between Mod1 and Mod2 , Mod1 and Mod3, and Mod2 and Mod3 are 2, 3, and 1 respectively. Thus, none of the model comparisons yield significant differences in model fit, so Mod1 is retained as the most parsimonious model with acceptable fit.

	-LL	Mod1 vs:	Mod2 vs:
Mod1	4025.265		
Mod2	4025.094	0.342	
Mod3	4024.137	2.256*	1.914

* This is similar to the corresponding SEM chi-square 2.250 from Mplus.

The Mod1 estimates closely match their counterparts from Mplus. The variance ratio estimate for the first two moderator categories is $(\exp(.026-.367))^2 = 0.5056$, agreeing with the Mplus estimate of 0.506. The variance ratio for the last two categories is $(\exp(.026-.367-.021-.508))^2 = 0.1755$, agreeing with the Mplus estimate of 0.176. The table below compares the remaining parameter estimates from SPSS and Mplus in each of the four moderator categories, demonstrating close agreement between the packages.

Categ.	Package	s_y	s_x	r
1	Mplus	1.441	1.025	.734
	SPSS	1.443	1.026	.734
2	Mplus	1.731	1.231	.734
	SPSS	1.732	1.231	.734
3	Mplus	2.395	1.005	.833
	SPSS	2.399	1.005	.833
4	Mplus	3.321	1.393	.495
	SPSS	3.330	1.395	.496

Reference

Smithson, M. (2012) *A simple statistic for comparing moderation of slopes and correlations*.